# Phylogenetic Inference using RevBayes

Substitution Models for Unconstrained Trees

Sebastian Höhna, Michael Landis, Tracy Heath and Brian Moore

#### 1 Overview

This tutorial demonstrates how to set up and perform analyses using common nucleotide substitution models. The substitution models used in molecular evolution are continuous time Markov models, which are fully characterized by their instantaneous-rate matrix:

$$Q = \begin{pmatrix} -\mu_A & \mu_{GA} & \mu_{CA} & \mu_{TA} \\ \mu_{AG} & -\mu_{G} & \mu_{CG} & \mu_{TG} \\ \mu_{AC} & \mu_{GC} & -\mu_{C} & \mu_{TC} \\ \mu_{AT} & \mu_{GT} & \mu_{CT} & -\mu_{T} \end{pmatrix} ,$$

where  $\mu_{ij}$  represents the instantaneous rate of substitution from state i to state j. Given the instantaneous-rate matrix, Q, we can compute the corresponding transition probabilities for a branch of length  $\nu$ ,  $P(\nu)$ , by exponentiating the rate matrix:

$$P(\nu) = \begin{pmatrix} p_{AA}(\nu) & p_{GA}(\nu) & p_{CA}(\nu) & p_{TA}(\nu) \\ p_{AG}(\nu) & p_{GG}(\nu) & p_{CG}(\nu) & p_{TG}(\nu) \\ p_{AC}(\nu) & p_{GC}(\nu) & p_{CC}(\nu) & p_{TC}(\nu) \\ p_{AT}(\nu) & p_{GT}(\nu) & p_{CT}(\nu) & p_{TT}(\nu) \end{pmatrix} = e^{Q\nu} = \sum_{j=0}^{\infty} \frac{\nu Q^j}{j!} .$$

Each specific substitution model has a uniquely defined instantaneous-rate matrix, Q.

In this tutorial you will perform phylogeny inference under common models of DNA sequence evolution: JC, F81, HKY85, GTR, GTR+Gamma and GTR+Gamma+I. For all of these substitution models, you will perform an MCMC analysis to estimate phylogeny and other model parameters. The estimated trees will be unrooted trees with independent branch-length parameters. All the assumptions will be covered more in detail later in this tutorial.

### Requirements

This tutorial that you have had an introduction to RevBayes, the Rev language, and MCMC. You can get this from the following tutorials:

- Getting Started with RevBayes
- Basic Introduction to Rev & MCMC

Note that the Basic Introduction introduces the basic syntax of Rev but does not cover any phylogenetic models. You may skip the this tutorial if you have some familiarity with R. We tried to keep this tutorial very basic and introduce all the language concepts on the way. You can refer to the Basic Introduction for a more in-depth discussion of concepts in Rev.

## 2 Data and files

We provide the data file of DNA sequences required for this tutorial. You may want to use your own data instead.

- → Create a folder called **data** and download the following files:
  - primates\_cytb.nex: Alignment of the cytochrome b subunit from 23 primates representing 14 of the 16 families (Indriidae and Callitrichidae are missing).

# 3 Example: Character Evolution under the Jukes-Cantor Substitution Model

## 3.1 Getting Started

The first section of this exercise involves: (1) setting up a Jukes-Cantor (JC) substitution model for an alignment of the cytochrome b subunit; (2) approximating the posterior probability of the tree topology and node ages (and all other parameters) using MCMC, and; (3) summarizing the MCMC output by computing the maximum a posteriori tree.

The general structure of the model is represented in Figure 1. We first consider the simplest substitution model described by Jukes and Cantor (1969). The instantaneous-rate matrix for the JC substitution model is defined as

$$Q_{JC69} = \begin{pmatrix} * & \frac{1}{3} & \frac{1}{3} & \frac{1}{3} \\ \frac{1}{3} & * & \frac{1}{3} & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & * & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & \frac{1}{3} & * \end{pmatrix} ,$$

which has the advantage that the transition probability matrix can be computed analytically:

$$P_{JC69} = \begin{pmatrix} \frac{1}{4} + \frac{3}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} \\ \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} + \frac{3}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} \\ \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} + \frac{3}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} \\ \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} - \frac{1}{4}e^{-rt} & \frac{1}{4} + \frac{3}{4}e^{-rt} \end{pmatrix},$$

where t is the branch length in units of time, and r is the rate (clock) for the process. In the later exercises you will be asked to specify more complex substitution models. It may seem like a lot of calculations, but the specification and computation of the rate matrix are done by RevBayes. Here we only provide some of the equations for the models in case you might be interested in the details. You will be able to complete the exercises without delving too deeply into the underlying math.




Figure 1: Graphical model representation of a simple phylogenetic model. The graphical model shows the dependencies between the parameters. Here, the rate matrix Q is a constant variable because it is fixed and does not depend on any parameters. The only free parameters of this model, the unconstrained Jukes-Cantor model, are the tree topology  $\Psi$  and the branch lengths  $\nu_i$ . The deterministic variable  $\mathbb L$  is the total tree length, which is computed by summing all of the branch lengths.

→ It is important to note that you can run all of the Rev commands from a RevBayes script file using the source() function in the RevBayes console. If you create a file called JukesCantor.Rev and save all of the commands listed below in this file, run the analysis by calling this function:

#### source("JukesCantor.Rev")

You can try this now if you run the file provided on line called <code>JukesCantor.nonclock.Rev</code>. If everything loaded properly, then you should see the program initiate the Markov chain Monte Carlo (MCMC) analysis that estimates the posterior distribution. If you continue to let this run, then you will see it output the states of the Markov chain once the MCMC analysis begins.

Ultimately, this is how you will execute most analyses in RevBayes, with the full specification of the model and analyses contained in the sourced files. You could easily run this entire analysis on your own data by substituting your data file name for that in the model-specification file. However, it is important to understand the components of the model to be able to take full advantage of the flexibility and richness of RevBayes. Furthermore, without inspecting the Rev scripts sourced in a file, you may end up inadvertently performing inappropriate analyses on your dataset, which would be a waste of your time and CPU cycles. The next steps will walk you through the full specification of the model and MCMC analyses.

## 3.2 Loading the Data

→ You should have already downloaded the data files in Section 2. Links to additional files, including the scripts to run these analyses can be found on the RevBayes tutorials website. Remember that the data file should be in a directory called data that is in your current working directory.

First load in the sequences using the readDiscreteCharacterData() function.

```
data <- readDiscreteCharacterData("data/primates_cytb.nex")</pre>
```

Executing these lines initializes the data matrix and the respective Rev variables. To report the current value of any variable, simply type the variable name and press enter. For the data matrix, this provides information about the alignment:

Next we will specify some useful variables based on our dataset. The variable data has member functions that we can use to retrieve information about the dataset. These include the number of species (n\_species), the tip labels (names), and the number of internal branches (n\_branches). Each of these variables will be necessary for setting up different parts of our model.

```
n_species <- data.ntaxa()
names <- data.names()
n_branches <- 2 * n_species - 3</pre>
```

Additionally, we'll set up a counter variable that counts the number of moves as they are added to the analysis. (Recall that moves are algorithms used to propose new parameter values during MCMC simulation.) This will make it much easier if we extend the model or analysis to include additional moves or to remove some moves.

```
mi = 0
```

You may have noticed that we used the = operator to create the move index. This simply means that the variable is used within the workspace and not part of the model. This operator will appear again later on, e.g., when we create moves and monitors.

With the data loaded, we can now proceed to specify our Jukes-Cantor substitution model.

#### 3.3 Jukes-Cantor Substitution Model

A given substitution model is defined by its corresponding instantaneous-rate matrix, Q. The Jukes-Cantor substitution model does not have any free parameters (as the substitution rates are all assumed to be equal), so we can define it as a constant variable. The function fnJC(n) will create an instantaneous-rate matrix for character with n states. Since we use DNA data here, we create a 4x4 instantaneous-rate matrix:

```
Q <- fnJC(4)
```

You can see the rates of the Q matrix by typing

```
Q
[[-1.0000, 0.3333, 0.3333],
0.3333, -1.0000, 0.3333],
0.3333, 0.3333, -1.0000, 0.3333],
0.3333, 0.3333, 0.3333, -1.0000]]
```

As you can see, all substitution rates are equal.

#### 3.4 Tree Topology and Branch Lengths

The tree topology and branch lengths are stochastic nodes in our phylogenetic model. In Figure 1, the tree topology is denoted  $\Psi$  and the length of the branch leading to node i is  $\nu_i$ .

We will assume that all possible labeled, unrooted tree topologies have equal probability. This is the dnUniformTopology() distribution in RevBayes. Specify the topology stochastic node by passing in the tip labels names to the dnUniformTopology() distribution:

```
topology ~ dnUniformTopology(names)
```

Some types of stochastic nodes can be updated by a number of alternative moves. Different moves may explore parameter space in different ways, and it is possible to use multiple different moves for a given parameter to improve mixing (the efficiency of the MCMC simulation). In the case of our unrooted tree topology, for example, we can use both a nearest-neighbor interchange move (mvNNI) and a subtree-prune and regrafting move (mvSPR). These moves do not have tuning parameters associated with them, thus you only need to pass in the topology node and proposal weight.

```
moves[++mi] = mvNNI(topology, weight=1.0)
moves[++mi] = mvSPR(topology, weight=1.0)
```

The weight specifies how often the move will be applied either on average per iteration or relative to all other moves. Have a look at the MCMC Diagnosis tutorial for more details about moves and MCMC strategies (found on the RevBayes Tutorials Website).

Next we have to create a stochastic node for each of the 2N-3 branches in our tree (where  $N=n\_species$ ). We can do this using a **for** loop — this is a plate in our graphical model. In this loop, we can create each of the branch-length nodes and assign each move. Copy this entire block of Rev code into the console:

```
for (i in 1:n_branches) {
   br_lens[i] ~ dnExponential(10.0)
   moves[++mi] = mvScale(br_lens[i])
}
```

It is convenient for monitoring purposes to add the tree length as deterministic variable. The tree length is simply the sum of all branch lengths. This parameter is denoted  $\mathbb{L}$  in Figure 1. Accordingly, the tree length can be computed using the  $\operatorname{sum}()$  function, which calculates the sum of any vector of values.

```
TL := sum(br_lens)
```

Finally, we can create a *phylogram* (a phylogeny in which the branch lengths are proportional to the expected number of substitutions/site) by combining the tree topology and branch lengths. We do this using the **treeAssembly()** function, which applies the value of the  $i^{th}$  member of the **br\_lens** vector to the branch leading to the  $i^{th}$  node in **topology**. Thus, the **phylogeny** variable is a deterministic node:

```
phylogeny := treeAssembly(topology, br_lens)
```

## 3.5 Putting it All Together

We have fully specified all of the parameters of our phylogenetic model—the tree topology with branch lengths, and the substitution model describing how the sequence data evolved over the tree with branch lengths. Collectively, these parameters comprise a distribution called the *phylogenetic continuous-time Markov chain*, and we use the **dnPhyloCTMC** constructor function to create this node. This distribution requires several input arguments: (1) the **tree** with branch lengths; (2) the instantaneous-rate matrix  $\mathbb{Q}$ , and; (3) the **type** of character data.

```
seq ~ dnPhyloCTMC(tree=phylogeny, Q=Q, type="DNA")
```

Once the dnPhyloCTMC model has been created, we can attach our sequence data to the tip nodes in the tree

```
seq.clamp(data)
```

Note that although we assume that our sequence data are random variables since they are realizations of our phylogenetic model, fix them to the values we observed as "clamped" nodes for the purposes of inference. When the clamp function is called, RevBayes sets each of the stochastic nodes representing the tips of the tree to the corresponding nucleotide sequence in the alignment. This essentially tells the program that we have observed data for the sequences at the tips.

Finally, we wrap the entire model to provide convenient access to the DAG. To do this, we only need to give the model() function a single node. With this node, the model() function can find all of the other nodes by following the arrows in the graphical model:

```
mymodel = model(Q)
```

Now we have specified a simple phylogenetic analysis—each parameter of the model will be estimated from every site in our alignment. If we inspect the contents of **mymodel** we can review all of the nodes in the DAG:

```
mymodel
```

#### 3.6 Performing an MCMC Analysis Under the Jukes-Cantor Model

In this section, will describe how to set up the MCMC sampler and summarize the resulting posterior distribution of trees.

#### 3.6.1 Specifying Monitors

For our MCMC analysis, we need to set up a vector of *monitors* to record the states of our Markov chain. The monitor functions are all called **mn\***, where **\*** is the wildcard representing the monitor type. First, we will initialize the model monitor using the **mnModel** function. This creates a new monitor variable that will output the states for all model parameters when passed into a MCMC function.

```
monitors[1] = mnModel(filename="output/primates_cytb_JC_posterior.log",printgen=10,
    separator = TAB)
```

The **mnFile** monitor will record the states for only the parameters passed in as arguments. We use this monitor to specify the output for our sampled trees and branch lengths.

```
monitors[2] = mnFile(filename="output/primates_cytb_JC_posterior.trees",printgen=10,
    separator = TAB, phylogeny)
```

Finally, create a screen monitor that will report the states of specified variables to the screen with mnScreen:

```
monitors[3] = mnScreen(printgen=1000, TL)
```

#### 3.6.2 Initializing and Running the MCMC Simulation

With a fully specified model, a set of monitors, and a set of moves, we can now set up the MCMC algorithm that will sample parameter values in proportion to their posterior probability. The mcmc() function will create our MCMC object:

```
mymcmc = mcmc(mymodel, monitors, moves)
```

We may wish to run the .burnin() member function. It is worth noting that this function does not specify the number of states that we wish to discard from the MCMC analysis as burnin (i.e., the samples collected before the chain converges to the stationary distribution). Instead, the .burnin() function specifies a *completely separate* preliminary MCMC analysis that is used to tune the scale of the moves to improve mixing of the MCMC analysis and the states are not written to file.

```
mymcmc.burnin(generations=10000,tuningInterval=1000)
```

Now, run the MCMC:

```
mymcmc.run(generations=30000)
```

When the analysis is complete, you will have the monitored files in your output directory.

Methods for visualizing the marginal densities of parameter values are not currently available in RevBayes itself. Thus, it is important to use programs like Tracer (Rambaut and Drummond 2011) to evaluate mixing and non-convergence. (RevBayes does, however, have a tool for convergence assessment called beca.)

→ Look at the file called output/primates\_cytb\_JC\_posterior.log in Tracer.

#### 3.7 Exercise 1

We are interested in the phylogenetic relationship of the Tarsiers. Therefore, we need to summarize the trees sampled from the posterior distribution. RevBayes can summarize the sampled trees by reading in the tree-trace file:

The mapTree() function will summarize the tree samples and write the maximum a posteriori tree to file:

```
mapTree(treetrace, "output/primates_cytb_JC.tree")
```

Fill in the following table as you go through the tutorial.

 $\rightarrow$  Look at the file called output/primates\_cytb\_JC.tree in FigTree.

Table 1: Posterior probabilities of phylogenetic relationship\*.

Model	Lemuroidea	Lorisoidea	Platyrrhini	Catarrhini	other
Jukes-Cantor					
HKY85					
F81					
GTR					
GTR+Γ					
$GTR+\Gamma+I$					
Your model 1					
Your model 2					•
Your model 3					
*vou can edit this table					

<sup>\*</sup>you can edit this table

Table 2: Primate species and famaly relationships.

Species	Family	Parvorder	Suborder
Alouatta palliata	Atelidae	Platyrrhini (NWM)	Haplorrhini
Aotus trivirgatus	Aotidae	Platyrrhini (NWM)	Haplorrhini
Callicebus donacophilus	Pitheciidae	Platyrrhini (NWM)	Haplorrhini
Cebus albifrons	Cebidae	Platyrrhini (NWM)	Haplorrhini
Cheirogaleus major	Cheirogaleidae	Lemuroidea	Strepsirrhini
Chlorocebus aethiops	Cercopithecoidea	Catarrhini	Haplorrhini
Colobus guereza	Cercopithecoidea	Catarrhini	Haplorrhini
Daubentonia madagascariensis	Daubentoniidae	Lemuroidea	Strepsirrhini
Galago senegalensis	Galagidae	Lorisidae	Strepsirrhini
Hylobates lar	Hylobatidea	Catarrhini	Haplorrhini
Lemur catta	Lemuridae	Lemuroidea	Strepsirrhini
Lepilemur hubbardorum	Lepilemuridae	Lemuroidea	Strepsirrhini
Loris tardigradus	Lorisidae	Lorisidae	Strepsirrhini
Macaca mulatta	Cercopithecoidea	Catarrhini	Haplorrhini
Microcebus murinus	Cheirogaleidae	Lemuroidea	Strepsirrhini
Nycticebus coucang	Lorisidae	Lorisidae	Strepsirrhini
Otolemur crassicaudatus	Galagidae	Lorisidae	Strepsirrhini
Pan paniscus	Hominoidea	Catarrhini	Haplorrhini
Perodicticus potto	Lorisidae	Lorisidae	Strepsirrhini
Propithecus coquereli	Indriidae	Lemuroidea	Strepsirrhini
Saimiri sciureus	Cebidae	Platyrrhini (NWM)	Haplorrhini
Tarsius syrichta	Tarsiidae		Haplorrhini
Varecia variegata variegata	Lemuridae	Lemuroidea	Strepsirrhini

# 4 The Hasegawa-Kishino-Yano (HKY) 1985 Substitution Model

The Jukes-Cantor model assumes that all substitution rates are equal, which also implies that the stationary frequencies of the four nucleotide bases are equal. These assumptions are not very biologically reasonable, so we might wish to consider a more realistic substitution model that relaxes some of these assumptions. For example, we might allow stationary frequencies,  $\pi$ , to be unequal, and allow rates of transition and transversion substitutions to differ,  $\kappa$ . This corresponds to the substitution model proposed by Hasegawa et al. (1985; HKY), which is specified with the following instantaneous-rate matrix:

$$Q_{HKY} = \begin{pmatrix} \cdot & \pi_C & \kappa \pi_G & \pi_T \\ \pi_A & \cdot & \pi_C & \kappa \pi_T \\ \kappa \pi_A & \pi_C & \cdot & \pi_T \\ \pi_A & \kappa \pi_C & \pi_G & \cdot \end{pmatrix} .$$

The diagonal  $\cdot$  entries are equal to the negative sum of the elements in the corresponding row.

→ Use the file JukesCantor.nonclock.Rev (containing all of the commands from Example 3) as a starting point for the HKY analysis.

Note that we are adding two new variables to our model. We can define a variable **pi** for the stationary frequencies that are drawn from a flat Dirichlet distribution by

```
pi_prior <- v(1,1,1,1)
pi ~ dnDirichlet(pi_prior)</pre>
```

Since **pi** is a stochastic variable, we need to specify a move to propose updates to it. A good move on variables drawn from a Dirichlet distribution is the **mvSimplexElementScale**. This move randomly takes an element from the simplex, proposes a new value for it drawn from a Beta distribution, and then rescales all values of the simplex to sum to 1 again.

```
moves[++mi] = mvSimplexElementScale(pi)
```

The second new variable is  $\kappa$ , which specifies the ratio of transition-transversion rates. The  $\kappa$  parameter must be a positive-real number and a natural choice as the prior distribution is the lognormal distribution:

```
kappa ~ dnLnorm(0.0,1.25)
```

Again, we need to specify a move for this new stochastic variable. A simple scaling move should do the job.

```
moves[++mi] = mvScale(kappa)
```

Finally, we need to create the HKY instantaneous-rate matrix using the fnHKY function:

```
Q := fnHKY(kappa,pi)
```

This should be all for the HKY model.

 $\rightarrow$  Don't forget to change the output file names, otherwise your old analyses files will be overwritten.

### 4.1 Exercise 2

- Copy the file called <code>JukesCantor.nonclock.Rev</code> and modify it by including the necessary parameters to specify the HKY substitution model (change the file name to something like <code>HKY.nonclock.Rev</code>).
- Run an MCMC analysis to estimate the posterior distribution under the HKY substitution model.
- Are the resulting estimates of the base frequencies equal? If not, how much do they differ? Are the estimated base frequencies similar to the empirical base frequencies? The empirical base frequencies are the frequencies of the characters in the alignment, which can be computed with RevBayes by data.getEmpiricalBaseFrequencies().

- Is the inferred rate of transition substitutions higher than the rate of transversion substitutions? If so, by how much?
- Like the HKY model, the Felsenstein 1981 (F81) substitution model has unequal stationary frequencies, but it assumes equal transition-transversion rates (Felsenstein 1981). Can you set up the F81 model and run an analysis?
- Complete the table of the phylogenetic relationship of Tarsiers.

## 5 The General Time-Reversible (GTR) Substitution Model

The HKY substitution model can accommodate unequal base frequencies and different rates of transition and transversion substitutions. Despite these extensions, the HKY model may still be too simplistic for many real datasets. Here, we extend the HKY model to specify the General Time Reversible (GTR) substitution model (Tavaré 1986), which allows all six exchangeability rates to differ (Figure 2).

The instantaneous-rate matrix for the GTR substitution model is:

$$Q_{GTR} = \begin{pmatrix} \cdot & r_{AC}\pi_C & r_{AG}\pi_G & r_{AT}\pi_T \\ r_{AC}\pi_A & \cdot & r_{CG}\pi_G & r_{CT}\pi_T \\ r_{AC}\pi_A & r_{CG}\pi_C & \cdot & r_{GT}\pi_T \\ r_{AC}\pi_A & r_{CT}\pi_C & r_{GT}\pi_G & \cdot \end{pmatrix} ,$$

where the six exchangeability parameters,  $r_{ij}$ , specify the relative rates of change between states i and j.

#### 5.1 Exchangeability Rate Parameters

The GTR model requires that we define and specify a prior on the six exchangeability rates, which we will describe using a flat Dirichlet distribution. As we did previously for the Dirichlet prior on base frequencies, we first define a constant node specifying the vector of concentration-parameter values using the  $\mathbf{v}()$  function:

```
er_prior <- v(1,1,1,1,1)
```

This node defines the concentration-parameter values of the Dirichlet prior distribution on the exchangeability rates. Now, we can create a stochastic node for the exchangeability rates using the **dnDirichlet()** function, which takes the vector of concentration-parameter values as an argument and the  $\sim$  operator. Together, these create a stochastic node named **er** ( $\theta$  in Figure 2):

```
er ~ dnDirichlet(er_prior)
```

The Dirichlet distribution assigns probability densities to a group of parameters: e.g., those that measure proportions and must sum to 1. Here, we have specified a six-parameter Dirichlet prior, where each value describes one of the six relative rates of the GTR model: (1) A = C; (2) A = G; (3) A = T; (4) C = G;




Figure 2: Graphical model representation of the General Time Reversible (GTR) phylogenetic model.

(5)  $C \hookrightarrow T$ ; (6)  $G \hookrightarrow T$ . The input parameters of a Dirichlet distribution are called shape (or concentration) parameters. The expectation and variance for each variable are related to the sum of the shape parameters. The prior we specified above is a 'flat' or symmetric Dirichlet distribution; all of the shape parameters are equal (1,1,1,1,1,1). This describes a model that allows for equal rates of change between nucleotides, such that the expected rate for each is equal to  $\frac{1}{6}$  (Figure 3a). We might also parameterize the Dirichlet distribution such that all of the shape parameters were equal to 100, which would also specify a prior with an expectation of equal exchangeability rates (Figure 3b). However, by increasing the values of the shape parameters, er\_prior <- v(100,100,100,100,100), the Dirichlet distribution will more strongly favor equal exchangeability rates; (i.e., providing is a relatively informative prior). Alternatively, we might consider an asymmetric Dirichlet parameterization that could reflect a strong prior belief that transition and transversion substitutions occur at different rates. For example, we might specify the prior density er\_prior <- v(4,8,4,4,8,4). Under this model, the expected rate for transversions would be  $\frac{4}{32}$  and that for transitions would be  $\frac{8}{32}$ , and there would be greater prior probability on sets of GTR rates that matched this configuration (Figure 3c). Yet another aymmetric prior could specify that each of the six GTR rates had a different value conforming to a Dirichlet (2,4,6,8,10,12). This would lead to a different prior probability density for each rate parameter (Figure 3d). Without strong prior knowledge about the pattern of relative rates, however, we can better reflect our uncertainty by using a vague prior on the GTR rates. Notably, all patterns of relative rates have the same probability density under er prior <v(1,1,1,1,1,1).

For each stochastic node in our model, we must also specify a proposal mechanism if we wish to estimate




Figure 3: Four different examples of Dirichlet priors on exchangeability rates.

that parameter. The Dirichlet prior on our parameter **er** creates a *simplex* of values that sum to 1.

```
moves[++mi] = mvSimplexElementScale(er)
```

We can use the same type of distribution as a prior on the 4 stationary frequencies  $(\pi_A, \pi_C, \pi_G, \pi_T)$  since these parameters also represent proportions. Specify a flat Dirichlet prior density on the base frequencies:

```
pi_prior <- v(1,1,1,1)
pi ~ dnDirichlet(pi_prior)</pre>
```

The node pi represents the  $\pi$  node in Figure 2. Now add the simplex scale move on the stationary frequencies to the moves vector:

```
moves[++mi] = mvSimplexElementScale(pi)
```

We can finish setting up this part of the model by creating a deterministic node for the GTR instantaneous-rate matrix Q. The fnGTR() function takes a set of exchangeability rates and a set of base frequencies to compute the instantaneous-rate matrix used when calculating the likelihood of our model.

```
Q := fnGTR(er,pi)
```

#### 5.2 Execise 3

- Use one of your previous analysis files—either the <code>JukesCantor.nonclock.Rev</code> or <code>HKY.Rev</code>—to specify a GTR analysis in a new file called <code>GTR.Rev</code>. Adapt the old analysis to be performed under the GTR substitution model.
- Run an MCMC analysis to estimate the posterior distribution.
- Complete the table of the phylogenetic relationship of Tarsiers.

## 6 The Discrete Gamma Model of Among Site Rate Variation

Members of the GTR family of substitution models assume that rates are homogeneous across sites, an assumption that is often violated by real data. We can accommodate variation in substitution rate among sites (ASRV) by adopting the discrete-gamma model (Yang 1994). This model assumes that the substitution rate at each site is a random variable that is described by a discretized gamma distribution, which has two parameters: the shape parameter,  $\alpha$ , and the rate parameter,  $\beta$ . In order that we can interpret the branch lengths as the expected number of substitutions per site, this model assumes that the mean site rate is equal to 1. The mean of the gamma is equal to  $\alpha/\beta$ , so a mean-one gamma is specified by setting the two parameters to be equal,  $\alpha = \beta$ . This means that we can fully describe the gamma distribution with the single shape parameter,  $\alpha$ . The degree of among-site substitution rate variation is inversely proportional to the value of the  $\alpha$ -shape parameter. As the value of the  $\alpha$ -shape increases, the gamma distribution increasingly resembles a normal distribution with decreasing variance, which therefore corresponds to decreasing levels of ASRV (Figure 4). By contrast, when the value of the  $\alpha$ -shape parameter




Figure 4: The probability density of mean-one gamma-distributed rates for different values of the  $\alpha$ -shape parameter.

is < 1, the gamma distribution assumes a concave distribution that concentrates most of the prior density on low rates, but retains some prior mass on sites with very high rates, which therefore corresponds to high levels of ASRV (Figure 4). Note that, when  $\alpha = 1$ , the gamma distribution collapses to an exponential distribution with a rate parameter equal to  $\beta$ .

We typically lack prior knowledge regarding the degree of ASRV for a given alignment. Accordingly, rather than specifying a precise value of  $\alpha$ , we can instead estimate the value of the  $\alpha$ -shape parameter from the data. This requires that we specify a diffuse (relatively 'uninformative') prior on the  $\alpha$ -shape parameter. For this analysis, we will use an exponential distribution with a rate parameter, **shape\_prior**, equal to **0.05**. An exponential prior assigns non-zero probability on values of  $\alpha$  ranging from 0 to  $\infty$ . The rate parameter of an exponential distribution, often denoted  $\lambda$ , controls both the mean and variance of this distribution, such that the expected (or mean) value of  $\alpha$  is:  $\mathbb{E}[\alpha] = \frac{1}{\lambda}$ . Thus, if we set  $\lambda = 0.05$ , then  $\mathbb{E}[\alpha] = 20$ .

This approach for accommodating ASRV is another example of a hierarchical model (Figure 5). That is, variation in substitution rates across sites is addressed by applying a site-specific rate multiplier to each of the j sites,  $r_j$ . These rate-multipliers are drawn from a discrete, mean-one gamma distribution; the shape of this prior distribution (and the corresponding degree of ASRV) is governed by the  $\alpha$ -shape parameter. The  $\alpha$ -shape parameter, in turn, is treated as an exponentially distributed random variable. Finally, the shape of the exponential prior is governed by the rate parameter,  $\lambda$ , which is set to a fixed value.




Figure 5: Graphical model representation of the General Time Reversible (GTR) + Gamma phylogenetic model.

## 6.1 Setting up the Gamma Model in RevBayes

Create a constant node called **shape\_prior** for the rate parameter of the exponential prior on the gamma-shape parameter (this is represented as the constant  $\lambda$ -rate parameter in Figure 5):

shape\_prior <- 0.05

Then create a stochastic node called **alpha** with an exponential prior (this represents the stochastic node for the  $\alpha$ -shape parameter in Figure 5):

```
alpha ~ dnExponential(shape_prior)
```

The way the ASRV model is implemented involves discretizing the mean-one gamma distribution into a set number of rate categories, k. Thus, we can analytically marginalize over the uncertainty in the rate at each site. The likelihood of each site is averaged over the k rate categories, where the rate multiplier is the mean (or median) of each of the discrete k categories. To specify this, we need a deterministic node that is a vector that will hold the set of k rates drawn from the gamma distribution with k rate categories. The fnDiscretizeGamma() function returns this deterministic node and takes three arguments: the shape and rate of the gamma distribution and the number of categories. Since we want to discretize a mean-one gamma distribution, we can pass in alpha for both the shape and rate.

Initialize the gamma\_rates deterministic node vector using the fnDiscretizeGamma() function with 4 bins:

```
gamma_rates := fnDiscretizeGamma( alpha, alpha, 4 )
```

Note that here, by convention, we set k = 4. The random variable that controls the rate variation is the stochastic node **alpha**. We will apply a simple scale move to this parameter.

```
moves[++mi] = mvScale(alpha, weight=2.0)
```

Remember that you need to call the dnPhyloCTMC constructor to include the new site-rate parameter:

```
seq ~ dnPhyloCTMC(tree=phylogeny, Q=Q, siteRates=gamma_rates, type="DNA")
```

#### 6.2 Execise 4

Modify the previous GTR analysis to specify the GTR+Gamma model. Run an MCMC simulation to estimate the posterior distribution.

- Is there an impact on the estimated phylogeny compared with the previous analyses? Look at the MAP tree and the posterior probabilities of the clades.
- What is the estimated tree length? Is the estimate different to the previous analysis? What could cause this?
- Complete the table of the phylogenetic relationship of Tarsiers.

## 7 Modeling Invariable Sites

All of the substitution models described so far assume that the sequence data are potentially variable. That is, we assume that the sequence data are random variables; specifically, we assume that they are realizations of the specified dnPhyloCTMC distribution. However, some sites may not be free to vary—when the substitution rate of a site is zero, it is said to be *invariable*. Invariable sites are often confused with *invariant* sites—when each species exhibits the same state, it is said to be invariant. The concepts are related but distinct. If a site is truly invariable, it will necessarily give rise to an invariant site pattern, as such sites will always have a zero substitution rate. However, an invariant site pattern may be achieved via multiple substitutions that happen to end in the same state for every species.

Here we describe an extension to our phylogenetic model to accommodate invariable sites. Under the invariable-sites model (Hasegawa et al. 1985), each site is invariable with probability pinvar, and variable with probability 1-pinvar.

First, let's have a look at the data and see how many invariant sites we have:

```
data.getNumInvariantSites()
```

There seem to be a substantial number of invariant sites.

Now let's specify the invariable-sites model in RevBayes. We need to specify the prior probability that a site is invariable. A Beta distribution is a common choice for parameters representing probabilities.

```
pinvar ~ dnBeta(1,1)
```

The Beta(1,1) distribution is a flat prior distribution that specifies equal probability for all values between 0 and 1.

Then, as usual, we add a move to change this stochastic variable; we'll used a simple sliding window move.

```
moves[mi++] = mvSlide(pinvar)
```

Finally, that you need to call the dnPhyloCTMC constructor to include the newpinvar parameter:

```
seq ~ dnPhyloCTMC(tree=phylogeny, Q=Q, siteRates=gamma_rates, pInv=pinvar, type="DNA")
```

#### 7.1 Exercise 5

- Extend the GTR model to account for invariable sites and run an analysis.
- What is the estimated probability of invariable sites and how does it relate to the ratio of invariant sites to the total number of sites?

- Extend the GTR+Γ model to account for invariable sites and run an analysis.
- What is the estimated probability of invariable sites now?
- Complete the table of the phylogenetic relationship of Tarsiers.

Questions about this tutorial can be directed to:

- Tracy Heath (email: phylo@iastate.edu)
- Sebastian Höhna (email: sebastian.hoehna@gmail.com)
- Michael Landis (email: mlandis@berkeley.edu)
- Brian R. Moore (email: brianmoore@ucdavis.edu)

# References

Felsenstein, J. 1981. Evolutionary trees from DNA sequences: A maximum likelihood approach. Journal of Molecular Evolution 17:368–376.

Hasegawa, M., H. Kishino, and T. Yano. 1985. Dating of the human-ape splitting by a molecular clock of mitochondrial DNA. Journal of Molecular Evolution 22:160–174.

Jukes, T. and C. Cantor. 1969. Evolution of protein molecules. Mammalian Protein Metabolism 3:21–132.

Rambaut, A. and A. J. Drummond. 2011. Tracer v1.5. http://tree.bio.ed.ac.uk/software/tracer/.

Tavaré, S. 1986. Some probabilistic and statistical problems in the analysis of DNA sequences. Some Mathematical Questions in BiologyDNA Sequence Analysis 17:57–86.

Yang, Z. 1994. Maximum likelihood phylogenetic estimation from DNA sequences with variable rates over sites: Approximate methods. Journal of Molecular Evolution 39:306–314.

Version dated: July 22, 2015